

```

Algoritmo p4_6
  Imprimir "Este programa encripta un mensaje escrito en minúsculas (sin tildes) por teclado cambiando las letras por otra en el abeceda"
  Imprimir "Por favor introduzca la frase:"
  Leer frase
  N=Longitud(frase)
  error=Falso
  Para i=1 hasta N
    letra=Subcadena(frase,i,i)
    Si (letra>="A" y letra<="Z") o letra=="á" o letra=="é" o letra=="í" o letra=="ó" o letra=="ú" o letra=="ü"
      error=Verdadero
    FinSi
  FinPara
  Si !error
    Dimension frasenum[N]
    abc="abcdefghijklmnopqrstuvwxyz"
    M=Longitud(abc)
    space="UVWXYZABCDEFGHIJKLMNÑOPQRST"
    Imprimir "Por favor introduzca la clave (número mayor a 1 menor que ",M,")"
    Leer clave
    Si clave<1 o clave>=M o clave-trunc(clave)!=0
      Imprimir "Error en la clave"
    Sino
      fraseCode=""
      Para i=1 hasta N
        break=Falso
        j=1
        Mientras (!break y j<=M)
          Si (Subcadena(frase,i,i)==Subcadena(abc,j,j))
            frasenum[i]=j
            break=Verdadero
          Sino
            Si Subcadena(frase,i,i)==" "
              frasenum[i]=0
              break=Verdadero
            Sino
              frasenum[i]=-1
            FinSi
          FinSi
          j=j+1
        FinMientras
        Si frasenum[i]==0
          fraseCode=Concatenar(fraseCode,Subcadena(space,clave,clave))
        Sino
          Si frasenum[i]==-1
            fraseCode=Concatenar(fraseCode,Subcadena(frase,i,i))
          Sino
            Si frasenum[i]+clave<=M
              fraseCode=Concatenar(fraseCode,Subcadena(abc,frasenum[i]+clave,frasenum[i]+clave))
            Sino
              fraseCode=Concatenar(fraseCode,Subcadena(abc,frasenum[i]+clave-M,frasenum[i]+clave-M))
            FinSi
          FinSi
        FinSi
      FinPara
      Imprimir "La frase queda encriptada como: ", fraseCode
    FinSi
  Sino
    Imprimir "Hay un error en la frase: ",frase,". Recuerde que debe está en minúsculas sin tildes"
  FinSi
FinAlgoritmo

```